

## MAPLE for CIRCUITS and SYSTEMS

E. L. Gerber, Ph.D

Drexel University

### ABSTRACT

There are three popular software programs used to solve circuits problems: Maple<sup>®</sup>, MATLAB<sup>®</sup>, and Spice<sup>®</sup>. Each one has a different approach to the manner in which the problem is implemented and solved. Spice will solve a circuit given its schematic diagram without the user having any knowledge of circuit analysis or circuit equations. MATLAB is useful in solving systems when the system is described in a strict and often unnatural format. This has the disadvantage of requiring the program input be written in a form that is removed from the mathematical representation of the real system. Maple solves equations in their basic form. That is the user must be able to write the appropriate equations for the system in order for Maple to solve.

Maple provides an extremely powerful “math-solving” computer package. Developed by mathematicians, it is not always user friendly in solving engineering problems. Maple is not designed to solve circuits or systems; however, it can solve most circuits and systems **equations**. In addition to being able to solve these equations quickly, it has substantial graphic capabilities. These two properties, speed and graphics, make it a valuable learning tool for electrical and systems engineers.

This paper will describe methods and procedures for using Maple to analyze and solve: signal analysis; first order and second order circuits; linear circuits; system transfer functions (Bode); and Laplace transform methods.

## **INTRODUCTION**

One of the most difficult aspects of an electrical engineering education is the application of mathematics to the analysis and study of electric circuits and systems. Even the better students require a long period of time and a lot of practice before they become proficient at solving these problems and develop insight to the material.

Many engineering schools use Maple in their Freshman math courses. Unfortunately the students are not required to use it again in other courses. Several years ago we decided to take advantage of the students' knowledge of Maple to include it in the first circuits and systems course. This is a sophomore class that deals with: signals; Kirchhoff's and Ohm's laws; first and second order linear circuits; Ac circuit analysis; phasors; Fourier series; transfer function analysis; and Bode plots. These topics require the student to have a strong mathematical background in order to perform well. Maple is the perfect tool for solving these problems. The solutions can easily be displayed to provide the student with a visual picture of the results in order to increase their ability to comprehend the physical significance of the results. A solution can easily be repeated with new parameters in order to demonstrate the effect these parameters have on the results. Often these can be displayed with several parameters varied on a single graph or in some cases a three dimensional graph.

## **CIRCUIT and SYSTEM EXAMPLES**

We will illustrate, with Maple, the following topics: signals; linear equations; first and second order circuits; transfer functions; and Laplace transforms. The Maple program used and the graphical results obtained will be shown for each example. All Maple commands must begin with a prompt, we will use the ( $\>$ ) symbol, and end with either a colon (:) or a semicolon (;). Several commands may be included in one line but the terminator must follow each. All the programs in this paper were performed on the student version of Maple V Release 4.

### **Signals**

First we will investigate the Maple procedure for defining and displaying electrical signals. Let's consider several common signals, and add them, multiply them, and determine some of their properties. We will use the "assign" definition in Maple which is the ( $:=$ ) notation. Dc, sinusoids, and an exponential function are defined below.


```
> S0 := 10:          S1 := 5*sin(2*t):  S2 := 6*sin(3*t):
      S3 := 6*sin(4*t+Pi/6):  E1 := exp(-2*t):
```

First we will add some of them and look at the results graphically. Then we will use Maple to calculate the period of a complex function and the maximum and minimum of a decaying sine function.

- Add S0 and S1, plot the results, and calculate the minimum value of the sum.

```
> S0 := 10, S1 := 5*sin(2*t): S01 := 10 + 5*sin(2*t):
> plot (S01, t=0..5);          minimize(S01);
                               5
```

The results, shown below, are simple and straightforward; the sine wave is shifted up by the Dc value. The minimum value of S01 (5) was calculated in a single Maple statement using the command “minimize”. The sum is plotted in Fig. 1.




**Fig. 1.  $S01 = 10 + 5\text{Sin}(2t)$**

- Add S2 and S3, and determine the period and then the frequency of the result.

```
> S23 := S2 + S3:
> plot (S23, t=0..5);
> Period :=fsolve(S23=0, t=4..5);
      Period := 4.413189680
> RadFreq :=evalf(2*Pi/Period);
```

```
RadFreq := 1.423728814
```


The graph (Fig.2) shows the first period starts at  $t = 0$  and ends near  $t = 4.4$  sec. We use the command “fsolve” to solve this expression for  $t$ , when  $S23 = 0$ , in the range from 4 to 5. The exact value of the period, to ten digits, is 4.413189680. This is defined as Period and is used to calculate the frequency in rad/sec.,  $\omega = 2\pi/T$ .



**Fig. 2.  $S23 = S2 + S3$**

- Next we will plot two sine waves, of different frequencies, parametrically. This will produce a Lissajous pattern. The Maple plot statement with square brackets [ ] will plot one function against the other. One sine wave is at 2 r/s and the other is at 3 r/s. The Lissajous pattern (Fig.3) shows that the ratio of the number of vertical peaks (6) to the number of horizontal peaks (4) is the same as the ratio of the two frequencies (3:2).


```
> plot( [S1,S2, t=0..8] );
```



**Fig. 3. Lissajous Graph of S1 and S2.**

- Now let's multiply the exponential term and the sine function,  $6\exp(-2t) * \sin(3t)$ , and plot the results. The result (Fig.4) has a peak value of about 2.5. To find the exact value we use the Maple command "maximize". We can also find the first zero-crossing time by solving the function  $SE = 0$  in the time range  $t = 0.5$  to  $1.5$ .

```
> SE := E1*S2:
> plot (SE, t= 0..3);
> evalf( maximize (SE) );
                2.592703785
> fsolve(SE= 0, t= 0.5..1.5);
                1.047197551
```




**Fig. 4.  $SE = 6e^{-2t}\sin 3t$**

- Next let's add two sine waves of the same frequency but different phases. Then we will use the animation capabilities of Maple to show the effect of varying the phase as the two functions are summed. The angle theta is defined as  $n*10^\circ$ , then n is incremented from 0 to 36. The graphs shown below are only two of the thirty-six parts of the animation display.

```
> S1 := sin(2*t):      Sn := sin(2*t + theta):
> theta := n*Pi/18:
> S1n := S1 + Sn:
```


```
> with (plots):
> animate (S1n, t = 0..3.6, n = 0..36);
```



**Fig.5.a) Phase shift • 180° b) Phase shift • 0°**

• Let’s consider piecewise functions now. Maple has a command “piecewise” that will create a piecewise function based on an input statement. The command format requires a *condition range* followed by the *function value*. For example, the unit step function, which is zero for  $t < 0$  and 1 for  $t > 0$ , is defined in Maple as follows,


```
> step := piecewise( t>0, 1, t<0, 0);
> plot(step, t = -2..4);
```



**Fig. 6. Unit Step**

A truncated ramp or a saw tooth, from  $t = 0$  to 2, is defined as,

```
> saw := piecewise (t<0, 0, t>0 and t<2, t);
      saw := { 0          t < 0
              t          -t < 0 and t - 2 < 0
            }
> plot (saw, t = -1..4);
```




**Fig. 7. One Saw Tooth**

In addition to the piecewise command, Maple has several defined singularity functions: the unit step (Heaviside) and the unit impulse (Dirac).

### Linear Systems

Electric circuits are solved by writing a number of linear simultaneous equations. Ohm's law is written for each resistor, Kirchhoff's current equations can be written at each node, and Kirchhoff's voltage equations can be written around each closed loop. This will generate a set of linear simultaneous equations describing the system. Maple can solve these equations in either of two forms. In the more basic form all possible circuit equations are written and Maple is requested to solve for all the variables. In the second, the more sophisticated, only the mesh or the node equations are written in matrix form and Maple can solve the matrix equation. Maple has a built-in package called "linear algebra" that contains about 100 linear algebra operations. Type "?linalg;" to see these operations.

- Let's consider the circuit below with three resistors and two voltage sources.




We will write three Ohm's equations for the three resistors, two KCL equations for the two nodes (b and e), and three KVL equations for three loops. This will provide us with a total of eight equations, more than we need. We will request Maple to solve these eight equations for the six unknown currents and voltages in this circuit.

1: $v1 = +8 \cdot I1$	abcdefa: $+v1 + v2 + V2 - V1 = 0$
2: $v2 = +10 \cdot I2$	abefa: $+v1 + v3 - V1 = 0$
3: $v3 = +1 \cdot I3$	bcdeb: $+v2 + V2 - v3 = 0$
b: $+i1 - i2 - i3 = 0$	e: $-i1 + i2 + i3 = 0$

The equations are assigned the names eq1 through eq8 and the responses are i1 through v3 in the Maple program below. Only one Maple command, "solve", is needed to solve the system of equations.

```
> eq1 := i1-i2-i3 = 0:      eq2 := -i1 +i2 +i3 = 0:
> eq3 := v1+v3-10= 0:      eq4 := v2 +12 -v3 = 0:
> eq5 := v1+v2 +12 -10 = 0: eq6 := v1= 8*i1:
> eq7 := v2 = 10*i2:      eq8 := v3 =1*i3:
> solutions :=solve( {eq1,eq2,eq3,eq4,eq5,eq6,eq7,eq8}, {i1,i2,i3,v1,v2,v3} );
      solutions := { i2 = -1, i1 = 1, v2 = -10, i3 = 2, v1 = 8, v3 = 2 }
```

Now consider the circuit shown below with three independent nodes, a, b, and c.



$$n = N - 1 = 4 - 1 = 3$$

Three node voltages:  
Y<sub>a</sub>, Y<sub>b</sub>, and Y<sub>c</sub>.

All conductances are 1 S.

The more sophisticated method of solving this circuit is to write the three node equations in matrix form and solve in Maple. The node equations of the circuit are written by inspection as,



$$\begin{aligned}
 Na : (1+1+1)V_a - (1)V_b - (1)V_c &= +9-3 \\
 Nb : -(1)V_a + (1+1+1)V_b - (1)V_c &= 3 \\
 Nc : -(1)V_a - (1)V_b + (1+1+1)V_c &= 7
 \end{aligned}$$

The matrix equation is also written by inspection for the circuit. We see that the [G] matrix is symmetrical about the major diagonal.

$$[G]V = [I_s] \quad \begin{vmatrix} +3 & -1 & -1 \\ -1 & +3 & -1 \\ -1 & -1 & +3 \end{vmatrix} \begin{vmatrix} V_a \\ V_b \\ V_c \end{vmatrix} = \begin{vmatrix} 6 \\ 3 \\ 7 \end{vmatrix}$$

We can have Maple solve for the three node voltages using matrix algebra. We define the [G] matrix in Maple using the “matrix” command, as shown below and then define the column matrix of the current sources using the “vector” command. Then we solve for the node voltage vector using the “linsolve” command. In addition, we will find the system determinate, the rank, inverse, and transpose of the [G] matrix.

```

> with(linalg):
> G := matrix(3,3, [3,-1,-1, -1,3,-1, -1,-1,3]);
      G :=  $\begin{bmatrix} 3 & -1 & -1 \\ -1 & 3 & -1 \\ -1 & -1 & 3 \end{bmatrix}$ 
> Is := vector([6,3,7]);
      Is :=  $[6 \ 3 \ 7]$ 
> V := linsolve(G,Is);
      V :=  $\begin{bmatrix} \frac{11}{2} & \frac{19}{4} & \frac{23}{4} \end{bmatrix}$ 
> det(G);
      16
> rank(G);

```


```

3
> inverse(G);
      [1/2  1/4  1/4]
      [1/4  1/2  1/4]
      [1/4  1/4  1/2]
> transpose(G);
      [3  -1  -1]
      [-1  3  -1]
      [-1  -1  3]

```

### First and Second Order Circuits

Consider the first order RL circuit below. The switch goes from position ‘a’ to ‘b’ at  $t = 0$ . The initial condition for the circuit is  $I_0 = 12V/2\Omega = 6$  A. After  $t = 0$  the voltage source  $v(t)$  is applied.



Let’s use Maple to solve for the current  $i(t)$  in the circuit. First we must write a Kirchhoff equation for the circuit for  $t > 0$ .

$$v(t) = Ri(t) + L\frac{di}{dt} = 2i(t) + 4\frac{di}{dt}$$

The Maple statement for differentiation wrt  $t$  is “diff( $i(t)$ ,  $t$ )”. Then we have Maple solve the differential equation, RL, using the command “dsolve”. Let’s first solve the homogeneous equation for  $i(t)$  with the initial condition  $i(0) = 6$  included in the brackets as follows,

```

> RL := 2*i(t) + 4*diff(i(t),t) = 0:
> sol_RL := dsolve ( {RL, i(0) = 6}, i(t) );
      sol_RL := i(t) = 6 e(-1/2 t)

```


The natural response of the circuit is an exponential decay function. Now consider a voltage source,  $v(t) = 34\sin(2t)$ , applied to the same circuit. The Maple statements with this source are rewritten as,

```
> RL_S := 2*i(t) + 4*diff(i(t),t) = 34*sin(2*t);
> solRL_S := dsolve({RL_S, i(0) = 6}, i(t) );
> simplify("");
```

$$i(t) = -4 \cos(2t) + \sin(2t) + 10 e^{\left(-\frac{1}{2}t\right)}$$

The solution has the decay term as before plus two terms due to the sine wave source. It is clear that at  $t = 0$  the current is  $(-4+0+10) = 6$ . The solutions of first order circuits are simple and straightforward in Maple. One defines the Kirchhoff equation of the circuit including the source, then uses the “dsolve” command with the initial conditions to obtain the complete solution to the circuit. This process can be repeated with other voltage sources applied, including piecewise functions defined earlier.

- Now consider the RLC circuit shown below with the voltage source and initial conditions known,  $i(0) = 5A$  and  $di(0)/dt = -3A/s$ .




The Kirchhoff voltage equation of this circuit, written below, is an integral-differential equation and it can be solved with Maple in one statement.

$$v_s(t) = Ri(t) + L\frac{di}{dt} + \frac{1}{C} \int_0^t i(t)dt = 3i(t) + 1\frac{di}{dt} + 2 \int_0^t i(t)dt$$

We will use a slightly different statement to solve; this solve command invokes the Laplace transform method of solving differential equations. First, we let  $v_s(t) = 12$  V, a Dc voltage. The Maple circuit equation and the solution statement for  $i(t)$ , including the initial conditions, are,

```
> RLC := 3*i(t) + diff(i(t),t) + 2*int(i(u), u=0..t) = 12:
> sol_DC := dsolve({RLC, i(0)=5, D(i)(0)=-3}, i(t), laplace);
      sol_DC := i(t) = -2 e(-2 t) + 7 e(-t)
> plot(rhs(sol_DC), t = 0..5);
```

The plot command requires a “rhs” statement in order to plot the solution of a differential equation (Fig.8).




**Fig. 8. Current in RLC Circuit with Dc Input.**

The solution is two decaying exponential functions and an initial value of 5.

- Next we will let  $v_s(t) = 78\sin(3t)$  in the same circuit and solve in a similar manner.

```
> RLC_sin := 3*i(t) + diff(i(t),t) + 2*int(i(u), u=0..t) = 78*sin(3*t):
> sol_sin := dsolve({RLC_sin, i(0)=5, D(i)(0)=-3}, i(t), laplace);
      sol_sin := i(t) = 46 e(-2 t) - 142/5 e(-t) - 63/5 cos(3 t) + 81/5 sin(3 t)
```



**Fig. 9 Current in RLC Circuit with Sine Wave Input**

The exponential terms (the natural response) are the same as for the Dc solution. However, the two sinusoidal terms are due to the driving source. The plot (Fig.9) shows how the exponential terms die after a few seconds and then the response continues at 3 rad/sec. We could easily continue applying other sources, including the piecewise functions described earlier, and follows the same procedure.

### **Transfer Functions**

As long as we are able to write the transfer function of a linear time-invariant system Maple should be able to investigate and graph the function. The function can be converted to the factored form or the polynomial form from an irrational form with Maple commands such as “factor” or “normal”. The poles and zeros, the limiting values, the root-locus diagram, and the frequency response characteristics of the function can be found easily in Maple. We will determine both the amplitude and phase response of a function written in terms of the complex frequency  $s$ . Linear or logarithmic graphs can be obtained. Consider the following example.

For the transfer function  $\mathbf{H(s)}$

$$H(s) = \frac{2500}{s^2 + 20s + 2500}$$

Find:

- |   |                                      |
|---|--------------------------------------|
| a) The poles and zeros                  | d) The Bode amplitude response in dB |
| b) The lower and upper frequency limits | e) The phase response                |
| c) The linear amplitude response        | f) The value of peak in dB.          |

a) First we define the function as  $H_S$  and solve for its poles and zeros. We use the Maple statement “denom” and “numer” to define the denominator and numerator of the function. The “solve” command will find the roots of the expression.

```
> H_S :=2500/(S^2+20*S+2500):
> poles :=solve(denom(H_S)):
> evalf(poles);
      -10.+48.98979486*I,
      -10.-48.98979486*I
> zeros :=solve(numer (H_S) );
```

Note,  $I = j = \sqrt{-1}$  in Maple. There is a pair of complex conjugate poles near  $-10 \pm j50$ . However, there is no answer to the “solve(numer)” statement. Maple gives no response when there is no solution to the command. In this situation it means that there are no zeros for this function.

b) Next we find the lower ( $S=0$ ) and the upper ( $S=+\infty$ ) frequency limits of the transfer function,  $H(S)$ .


```
> limit(H_S, S=0);
      1
> limit(H_S, S=+infinity);
      0
```

The lower limit value is 1 and the upper is 0, indicating that the function is a low pass filter.

c) Now we substitute  $j\omega$  for  $S$  in the transfer function and define the **magnitude** of  $H(j\omega)$  using the “abs” command. Then we will plot the magnitude of  $H(j\omega)$  vs. frequency from  $\omega = 0$  to 100 r/s, on linear scales. We see that the function is a low pass filter with a peak near  $\omega = 50$  r/s.

```
> H_W :=subs(S=I*W, H_S):
> Mag_H := abs(H_W):
```


```
> plot ((Mag_H), W=0..100);
```




**Fig. 10. Magnitude of H(jW) vs. W.**

d,e) Next we will obtain a Bode plot in dB vs. log frequency in one statement, and the phase plot in a second statement. The dB is defined as  $20 \cdot \log_{10}|H(jW)|$ .

```
> plots[semilogplot] (20*log[10](abs(H_W)), W=10..100);  
> plots[semilogplot]((180/Pi)*argument (H_W), W=10..500);
```



**Fig. 11. Bode Plot of H, dB vs. Log W.**



**Fig. 12. Phase Plot of  $H(jW)$  vs.  $\text{Log } W$ .**

f) The maximum value of the function (see Fig.11) can be evaluated in dB as follows,

```

> evalf(20*log[10] (maximize (Mag_H) ));
      8.136087846

```


- Three dimensional graphs and root-locus diagrams of transfer function are easily obtained from Maple. Let's look at a 3-d plot of the magnitude of  $H$ , on the z-axis, with the complex plane on the x- and y-axes. The Maple command is simply "plot3d", with  $S$  replaced by  $(x + jy)$ .

```

> H_S := 2500/(S^2 + 20*S + 2500):
> H_xy :=subs(S=x+I*y, H_S):
> plot3d(abs(H_xy), x=-30..10, y=-100..100,
      view=0..8, labels=[x,y,H]);

```






**Fig. 13. 3-D Plot of Magnitude  $H(jW)$  on Complex Plane.**

In Fig. 13 we see two poles located near  $x = -10$  and  $y = \pm j50$  on the complex plane. No zeros appear in this graph. These plots are in color.

### **Laplace Transforms**

Transform calculations (and inverse transforms) are easily implemented in Maple with the transform package called “intrans”. Consider the problem depicted below. The input voltage is a single square pulse of fundamental frequency 1/2 Hz or period of 2 seconds. It is defined in Maple as (SQRW) using the “piecewise” command. This input is applied to a third-order low pass Butterworth filter ( $H(S)$ ) with a corner frequency of 2 Hz and is defined as (B\_3).




The output voltage ( $V_o(t)$ ) of the system is desired. We can take the laplace transform of the input signal ( $V_{in\_S}$ ) and multiply it by the transfer function of the filter,  $H(S)$ . Then the output voltage  $V_o(t)$  is the inverse laplace transform of this product  $VO\_t$  in the program below.  $V_o(S) = H(S) \times V_{in}(S)$ . We will demon-strate this process in Maple as follows, and plot both the input and output signals.

```

> SQRW:=piecewise( t>0 and t<1,1, t>1 and t<2, -1):
> C :=2*Pi*2:
> B_3 := C^3/(S^3 + 2*C*S^2 + 2*C^2*S + C^3):
> with(intrans):    Vin_S :=laplace(SQRW,t,S):
> VO_S := (Vin_S)* (B_3):
> VO_t :=invlaplace(VO_S,S,t):
> plot([VO_t, SQRW], t=0..3);

```



**Fig. 14. Input and Output Signals of Third Order Filter.**

## SUMMARY

Two years ago we implemented a new curriculum in the ECE Department. As part of the change we require four two-credit laboratory courses in the third and fourth years of our five year Coop program. About one half of the 36 weekly experiments (9 per term) involve circuits or systems. We have made available to these students a new facility with 12 new computers with Maple, PSpice, and MATLAB. In most situations we allow the students to select the program they wish to use. We have found that most often they select Maple and do a lot more than what was required. We believe that Maple is the best choice for the students to learn this material.

## References:

**E. L. GERBER**, received the Ph.D. from University of Pennsylvania, is Professor of electrical engineering at Drexel University. Formerly Assistant Head of ECE Dept. at Drexel and director of undergraduate laboratories. Was Division Chair and Program Chair of Instrumentation Division of ASEE for almost a decade.